# Computation of the Wodzicki – Chern – Simons form in local coordinates. Computations for $S^1$ actions on $S^2 \times S^3$ (Last update : July 1, 2010.)

We define a local system of coordinates for $M = S^2 \times S^3$ and compute the Christoffel symbols and curvature. We use a system of coordinates to write a family of metrics over M taken from

J.P. Gauntlett, D. Martelli, J. Sparks and D. Waldram
Sasaki-Einstein metrics on $S^2 \times S^3$.
Adv. Theor. Math. Phys. 8 (2004) 711
[hep-th/0403002].

In this system of coordinates, M can be considered as a non-trivial $S^1$ fibration over $S^2 \times S^2$. We compute the zeroth and -1 order symbols for the connection form and the zeroth and -1 order symbol for the curvature form on the loop space LM. These computations allow us to calculate the relative Wodzicki-Chern-Simons form on LM. The natural $S^1$ action on the fibers of $M$ determines a homology class in LM. We compute the value of the WCS form on this class and related classes for the n-fold iteration of the $S^1$ action. Each of these integrals are different (i.e. determine nonhomologous classes in $H_5$(LM) ) for n = 0, 1, 2, .... As explained in the paper, this proves that the action is not smoothly homotopic to the trivial action, and that $\pi_1(\mathrm{Diff}(S^2 \times S^3))$ is infinite.

We proceed as follows: In Section 1, we compute the Christoffel symbols and curvature for these metrics, and use them to compute explicitly all needed symbols for the corresponding metrics on LM. In Section 2, we use these computations to calculate the Wodzicki-Chern-Simons form on LM. We integrate the WCS form over a 5-cycle in LM associated to a circle action on M. For certain metrics, the integral is nonzero.

**Table of contents**

1. Preliminaries -- the main procedures to compute Christoffel symbols,
curvature, and symbols for the connection and curvature forms on loop space.
1.1 Computation of the metric.
1.2 Computations of Christoffel symbols and curvature.
1.3 The symbols
2. Evaluation of the relative Wodzicki – Chern – Simons form on a cycle in $L(S^2 \times S^3)$
associated to the fiber action
2.1 The contribution from $\mathrm{Tr}\,\sigma_{-1}(\omega) \wedge \sigma_0(\Omega) \wedge \sigma_0(\Omega)$

# 1. Preliminaries

```
SetDirectory["Mathematica/PaperSteveYoshi"];
```

```
SetDirectory["research/fabian/07/notebooks"];
```

Define the metric in spherical coordinates on $S^2 \times S^3$. $\phi$ and $\theta$ are the ordinary spherical coordinates for $S^2$, and $\psi$, y and $\tau$ are coordinates for $S^3$. We have
$\phi$ in [0, $2\pi$], $\theta$ in [0, $\pi$]; with $\theta$ the angle with respect to the z axis.
$\psi$ in [0, 2 $\pi$], $\tau$ in [0, $2\pi$l], with l and the range of y defined below. In Gauntlett et al., the range for $\tau$ and y are chosen so the metric extends to the closure $S^2 \times S^3$ of the coordinate ball.

```
Clear[coord, metric, inversemetric,
  Christoffel, Theta, Phi, Xi, x, X1, X2, X3, X4, A, Con, NumIt];
dimension = 5;
coord = {Phi, Theta, Psi,  y, Tau};
PsiForm = {P1si, P2si, P3si, P4si};
```

```
XField = {X1, X2, X3, X4, X5};
YField = {Y1, Y2, Y3, Y4, Y5};
WField = {W1, W2, W3, W4, W5};
ZField = {Z1, Z2, Z3, Z4, Z5};
TField = {T1, T2, T3, T4, T5};
DotGamma = {Dot1, Dot2, Dot3, Dot4, Dot5};
```

## 1.1 Computation of the metric

Computation of the general metric: First, we introduce the line element from (2.1) in Gauntlett et al.

```
Clear[a, c]; coord1 = {Phi1, Theta1, Psi1, y1, Tau1};
Funcw[y_, a_, c_] := 2 * (a - y * y) / (1 - c * y);
Funcq[y_, a_, c_] := (a - 3 * y * y + 2 y * y * y * c) / (a - y * y);
```

Gauntlett's et al. line element $ds^2$. Here a and c are constants.

```
DeltaSquare[Phi1_, Theta1_, Psi1_, y1_, Tau1_, a_, c_] :=
  (1 - c * y) / 6 (Theta1 * Theta1 + Sin[Theta] * Sin[Theta] * Phi1 * Phi1) + (1 / Funcw[y, a, c]) *
    (1 / Funcq[y, a, c]) * y1 * y1 + 1 / 9 Funcq[y, a, c] * (Psi1 - Cos[Theta] * Phi1) ^2 +
   Funcw[y, a, c] * (Tau1 + (a * c - 2 * y + y * y * c) / (6 (a - y * y)) (Psi1 - Cos[Theta] * Phi1 )) ^2;
```

```
dimension = 5; Clear[a, c];
GenMetric[Phi_, Theta_, Psi_, y_, Tau_, a_, c_] :=
 Table[D[DeltaSquare[Phi1, Theta1, Psi1, y1, Tau1, a, c], coord1[[ii]], coord1[[jj]]] /
   If[ii == jj, 2, 2], {ii, 1, dimension}, {jj, 1, dimension}]
```

```
Simplify[GenMetric[Phi, Theta, Psi, y, Tau, a, c]]
```

$$\left\{\left\{-\frac{5 + a c^2 - 12 c y + 6 c^2 y^2 + \left(-1 + a c^2\right) \mathrm{Cos}[2\,\mathrm{Theta}]}{36\,(-1 + c y)}, 0, \frac{\left(2 + a c^2 - 6 c y + 3 c^2 y^2\right) \mathrm{Cos}[\mathrm{Theta}]}{18\,(-1 + c y)}, 0, \frac{(a c + y\,(-2 + c y))\,\mathrm{Cos}[\mathrm{Theta}]}{-3 + 3 c y}\right\}, \left\{0, \frac{1}{6}(1 - c y), 0, 0, 0\right\},\right.$$
$$\left\{\frac{\left(2 + a c^2 - 6 c y + 3 c^2 y^2\right) \mathrm{Cos}[\mathrm{Theta}]}{18\,(-1 + c y)}, 0, \frac{2 + a c^2 - 6 c y + 3 c^2 y^2}{18 - 18 c y}, 0, \frac{a c + y\,(-2 + c y)}{3 - 3 c y}\right\},$$
$$\left\{0, 0, 0, \frac{1 - c y}{2 a - 6 y^2 + 4 c y^3}, 0\right\},$$
$$\left.\left\{\frac{(a c + y\,(-2 + c y))\,\mathrm{Cos}[\mathrm{Theta}]}{-3 + 3 c y}, 0, \frac{a c + y\,(-2 + c y)}{3 - 3 c y}, 0, \frac{2\left(-a + y^2\right)}{-1 + c y}\right\}\right\}$$

Explicit expression for the metric $ds^2$ . We set c=1 as in Gauntlett et al.

```
c = 1; GenMetric[Phi, Theta, Psi, y, Tau, a, c]
```

$$\left\{\left\{\frac{1}{2}\left(\frac{(a-2y+y^2)^2\,\text{Cos}[\text{Theta}]^2}{9\,(1-y)\,(a-y^2)}+\frac{2\,(a-3y^2+2y^3)\,\text{Cos}[\text{Theta}]^2}{9\,(a-y^2)}+\frac{1}{3}\,(1-y)\,\text{Sin}[\text{Theta}]^2\right),\,0,\,\frac{1}{2}\left(-\frac{(a-2y+y^2)^2\,\text{Cos}[\text{Theta}]}{9\,(1-y)\,(a-y^2)}-\frac{2\,(a-3y^2+2y^3)\,\text{Cos}[\text{Theta}]}{9\,(a-y^2)}\right),\,0,\,-\frac{(a-2y+y^2)\,\text{Cos}[\text{Theta}]}{3\,(1-y)}\right\},\left\{0,\,\frac{1-y}{6},\,0,\,0,\,0\right\},\left\{\frac{1}{2}\left(-\frac{(a-2y+y^2)^2\,\text{Cos}[\text{Theta}]}{9\,(1-y)\,(a-y^2)}-\frac{2\,(a-3y^2+2y^3)\,\text{Cos}[\text{Theta}]}{9\,(a-y^2)}\right),\,0,\,\frac{1}{2}\left(\frac{(a-2y+y^2)^2}{9\,(1-y)\,(a-y^2)}+\frac{2\,(a-3y^2+2y^3)}{9\,(a-y^2)}\right),\,0,\,\frac{a-2y+y^2}{3\,(1-y)}\right\},\left\{0,\,0,\,0,\,\frac{1-y}{2\,(a-3y^2+2y^3)},\,0\right\},\left\{-\frac{(a-2y+y^2)\,\text{Cos}[\text{Theta}]}{3\,(1-y)},\,0,\,\frac{a-2y+y^2}{3\,(1-y)},\,0,\,\frac{2\,(a-y^2)}{1-y}\right\}\right\}$$

End of the computation of the metric.

## 1.2 Computations of Christoffel symbols and curvature

$$\textbf{metric} = \left\{\left\{\frac{1}{2}\left(\frac{(a-2y+y^2)^2\,\text{Cos}[\text{Theta}]^2}{9\,(1-y)\,(a-y^2)}+\frac{2\,(a-3y^2+2y^3)\,\text{Cos}[\text{Theta}]^2}{9\,(a-y^2)}+\frac{1}{3}\,(1-y)\,\text{Sin}[\text{Theta}]^2\right),\,0,\,\frac{1}{2}\left(-\frac{(a-2y+y^2)^2\,\text{Cos}[\text{Theta}]}{9\,(1-y)\,(a-y^2)}-\frac{2\,(a-3y^2+2y^3)\,\text{Cos}[\text{Theta}]}{9\,(a-y^2)}\right),\,0,\,-\frac{(a-2y+y^2)\,\text{Cos}[\text{Theta}]}{3\,(1-y)}\right\},\left\{0,\,\frac{1-y}{6},\,0,\,0,\,0\right\},\left\{\frac{1}{2}\left(-\frac{(a-2y+y^2)^2\,\text{Cos}[\text{Theta}]}{9\,(1-y)\,(a-y^2)}-\frac{2\,(a-3y^2+2y^3)\,\text{Cos}[\text{Theta}]}{9\,(a-y^2)}\right),\,0,\,\frac{1}{2}\left(\frac{(a-2y+y^2)^2}{9\,(1-y)\,(a-y^2)}+\frac{2\,(a-3y^2+2y^3)}{9\,(a-y^2)}\right),\,0,\,\frac{a-2y+y^2}{3\,(1-y)}\right\},\left\{0,\,0,\,0,\,\frac{1-y}{2\,(a-3y^2+2y^3)},\,0\right\},\left\{-\frac{(a-2y+y^2)\,\text{Cos}[\text{Theta}]}{3\,(1-y)},\,0,\,\frac{a-2y+y^2}{3\,(1-y)},\,0,\,\frac{2\,(a-y^2)}{1-y}\right\}\right\};$$

Computation of the metric inverse.

```
inversemetric = Simplify[Inverse[metric]];
```

```
metric = Simplify[metric];
inversemetric = Simplify[inversemetric];
```

Definition of the Christoffel Symbols. Notation: Christoffel[i,j,k]:=\Gamma^i_{j,k}.

```
Christoffel = Christoffel = Table[
    Simplify[(1 / 2) * Sum[(inversemetric[[ii, ss]]) * (D[metric[[ss, jj]], coord[[kk]]] +
          D[metric[[ss, kk]], coord[[jj]]] - D[metric[[jj, kk]], coord[[ss]]]),
       {ss, 1, dimension}]], {ii, 1, dimension}, {jj, 1, dimension}, {kk, 1, dimension}];
```

```
listaffine := Table[{ToString[Γ[i, j, k]], Christoffel[[i, j, k]]},
  {i, 1, dimension}, {j, 1, dimension}, {k, 1, dimension}]
```

Definition of the Curvature. Notation: Curvature[i,j,k,l]:=R^i_{jkl}.

```
Curvature = Curvature =
   Table[Simplify[D[Christoffel[[ii, jj, ll]], coord[[kk]]] - D[Christoffel[[ii, jj, kk]],
        coord[[ll]]] + Sum[Christoffel[[ss, jj, ll]] Christoffel[[ii, kk, ss]] -
          Christoffel[[ss, jj, kk]] Christoffel[[ii, ll, ss]], {ss, 1, dimension}]],
     {ii, 1, dimension}, {jj, 1, dimension}, {kk, 1, dimension}, {ll, 1, dimension}];
```

```
ListCurvature := Table[If[UnsameQ[Curvature[[i, j, k, l]], 0],
    {ToString[R[i, j, k, l]], Curvature[[i, j, k, l]]}],
  {i, 1, dimension}, {j, 1, dimension}, {k, 1, dimension}, {l, 1, dimension}]
TableForm[Partition[DeleteCases[Flatten[ListCurvature], Null], 2], TableSpacing → {2, 2}]
```

Computation of the covariant derivative for the curvature. Notation: CovCurvature[i,j,k,m,p]=R^{m}_{\ ijk;p}

```
CovCurvature = CovCurvature = Table[D[Curvature[[m, i, j, k]], coord[[p]]] +
      Sum[Curvature[[t, i, j, k]] * Christoffel[[m, p, t]], {t, 1, dimension}] -
      Sum[Curvature[[m, t, j, k]] * Christoffel[[t, p, i]], {t, 1, dimension}] -
      Sum[Curvature[[m, i, t, k]] * Christoffel[[t, p, j]], {t, 1, dimension}] -
      Sum[Curvature[[m, i, j, t]] * Christoffel[[t, p, k]], {t, 1, dimension}],
     {i, 1, dimension}, {j, 1, dimension}, {k, 1, dimension},
     {m, 1, dimension}, {p, 1, dimension}];
```

Computation of the Ricci curvature $Ricci_{mk} = R^{l}{}_{mlk}$. This step is not needed for the computations related to the WCS form; we just did this to confirm the metric is Einstein with Einstein constant 4.

```
Ricci = Table[Sum[Curvature[[kk, ii, kk, jj]], {kk, 1, dimension}],
    {ii, 1, dimension}, {jj, 1, dimension}];
```

```
Ricci2 = Table[
    Sum[D[Christoffel[[kk, j, i]], coord[[kk]]] - D[Christoffel[[kk, kk, i]], coord[[j]]] +
      Sum[Christoffel[[kk, kk, s]] Christoffel[[s, j, i]] -
        Christoffel[[s, i, kk]] Christoffel[[kk, j, s]], {s, 1, dimension}],
     {kk, 1, dimension}], {i, 1, dimension}, {j, 1, dimension}];
```

```
Simplify[Ricci2 - 4 * metric] // TableForm
```

| | | | | |
|---|---|---|---|---|
| 0 | 0 | 0 | 0 | 0 |
| 0 | 0 | 0 | 0 | 0 |
| 0 | 0 | 0 | 0 | 0 |
| 0 | 0 | 0 | 0 | 0 |
| 0 | 0 | 0 | 0 | 0 |

## 1.3 The symbols

Computation of the symbols for the connection.

```
SigmaMinusConn[XF_] :=
  -1 / 2 * Table[Sum[(-2 Curvature[[jj, kk, mm, ll]] - Curvature[[jj, mm, kk, ll]] +
         Curvature[[jj, ll, mm, kk]]) * XF[[mm]] * DotGamma[[ll]], {ll, 1, dimension},
      {mm, 1, dimension}] , {jj, 1, dimension}, {kk, 1, dimension}];
```

```
SigmaZeroConn[XF_] :=
  Table[Sum[Christoffel[[jj, mm, kk]] * XF[[mm]], {mm, 1, dimension}] ,
   {jj, 1, dimension}, {kk, 1, dimension}];
SigmaMinusConn[XF_] := Simplify[-1 / 2 * Table[
      Sum[(Curvature[[jj, ll, mm, kk]]) * XF[[mm]] * DotGamma[[ll]], {ll, 1, dimension},
       {mm, 1, dimension}] , {jj, 1, dimension}, {kk, 1, dimension}]];
```

Computation of the symbols for the curvature.

```
SigmaMinusCurOne[XF_, YF_] := 1 / 2 * Table[
    Sum[(2 * CovCurvature[[ kk, ll, nn, jj, mm]] + CovCurvature[[ ll, kk, nn, jj, mm]] +
         CovCurvature[[ nn, ll, kk, jj, mm]]) * XF[[mm]] * YF[[ll]] * DotGamma[[nn]],
       {nn, 1, dimension}, {mm, 1, dimension}, {ll, 1, dimension}] +
      Sum[YF[[ll]] * (2 * Curvature[[ jj, kk, ll, mm]] +
         Curvature[[ jj, ll, kk, mm]] + Curvature[[ jj, mm, ll, kk]]) *
        (Sum[D[XF[[mm]], coord[[nn]]] * DotGamma[[nn]], {nn, 1, dimension}] +
         Sum[Christoffel[[mm, ee, ff]] * DotGamma[[ee]] * XF[[ff]],
          {ee, 1, dimension}, {ff, 1, dimension}]), {mm, 1, dimension},
       {ll, 1, dimension}], {jj, 1, dimension}, {kk, 1, dimension}];
```

```
SigmaZeroCur[XF_, YF_] :=
  -Table[Sum[Curvature[[jj, kk, mm, ll]] * XF[[mm]] * YF[[ll]], {mm, 1, dimension},
     {ll, 1, dimension}] , {jj, 1, dimension}, {kk, 1, dimension}];
SigmaMinusCurOne[XF_, YF_] := 1 / 2 * Table[
    Sum[(2 * CovCurvature[[ kk, ll, nn, jj, mm]] + CovCurvature[[ ll, kk, nn, jj, mm]] +
         CovCurvature[[ nn, ll, kk, jj, mm]]) * XF[[mm]] * YF[[ll]] * DotGamma[[nn]],
      {nn, 1, dimension}, {mm, 1, dimension}, {ll, 1, dimension}] +
     Sum[YF[[ll]] * (2 * Curvature[[ jj, kk, ll, mm]] +
         Curvature[[ jj, ll, kk, mm]] + Curvature[[ jj, mm, ll, kk]]) *
        (Sum[D[XF[[mm]], coord[[nn]]] * DotGamma[[nn]], {nn, 1, dimension}] +
         Sum[Christoffel[[mm, ee, ff]] * DotGamma[[ee]] * XF[[ff]],
          {ee, 1, dimension}, {ff, 1, dimension}]), {mm, 1, dimension},
      {ll, 1, dimension}], {jj, 1, dimension}, {kk, 1, dimension}];
SigmaMinusCur[XF_, YF_] := SigmaMinusCurOne[XF, YF] - SigmaMinusCurOne[YF, XF];
```

# 2. Evaluation of the relative Wodzicki-Chern-Simons form on a cycle in LM associated to the fiber action.

As explained in the paper, the $S^1$action on the fiber determines a cycle in LM. We evaluate the WCS form on this cycle by pullback to an integral over $M$ and show that this integral is non-zero. This implies that the $S^1$ action is nontrivial in the sense defined in the paper.

We compute the WCS form at a typical loop $\gamma(\tau)$ of the $S^1$ action. In the coordinates from Gauntlett et al. the loop obtained is:

$S^1 \;{-}{>}\; L\left(S^2 \times S^3\right)$

t $\;\to (\phi, \theta, \psi$, y, $\tau$ + Palpha*t),

where Palpha measures the number of iterates of the action.

We compute the derivative of $\gamma$(t) with respect to t, and express the result in terms of the of the standard basis for $T_{\gamma(t)}\, S^2 \times S^3$. The outcome is stored in the variable (Dot1, Dot2, Dot3, Dot4, Dot5)=(0,0,0,0,Palpha).

This is an action via isometries, since the metric is independent of $\tau$. In the WCS integrand, the pullback is an integral over $S^1$ of "$\omega \wedge \Omega \wedge \Omega$" acting on $a_*(\text{partial}_\phi)(t)$, $a_*(\text{partial}_\psi)(t)$, ... at $a(t)$, which we call $\gamma(t)$. We store in the vector fields X, Y, Z, T, and W the pushforwards $a_*(\text{partial}_\phi)(t)$, $a_*(\text{partial}_\psi)(t)$, .... Since our action is an isometry, this integrand is independent of t, so instead of integrating over $S^1$, we can just multiply the integrand by $2\pi$. Below we first compute the integrand and then the pullback integral.

```
InitializeAction = Module[{Dotgam},
  Dot1 = 0;
  Dot2 = 0;
  Dot3 = 0;
  Dot4 = 0;
  Dot5 = PAlpha;
  (* The vector fields *)
  X1 = 1; X2 = 0; X3 = 0; X4 = 0; X5 = 0;
  Y1 = 0; Y2 = 1; Y3 = 0; Y4 = 0; Y5 = 0;
  Z1 = 0; Z2 = 0; Z3 = 1; Z4 = 0; Z5 = 0;
  T1 = 0; T2 = 0; T3 = 0; T4 = 1; T5 = 0;
  W1 = 0; W2 = 0; W3 = 0; W4 = 0; W5 = 1;
 ]
```

The next two matrices contain all the permutations of the fields X,Y,Z,W and T and their corresponding sign. They will be used to compute the evaluation of each permutation of the basis vector fields that contribute to the CSW form.

```
Permu = ListPermutations = Permutations[{XField, YField, ZField, WField, TField}];
SignPermu = {-1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1,
   -1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1,
   1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1,
   1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1,
   -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1, -1, 1, 1, -1};
```

## 2.1 The contribution from Tr $\sigma_{-1}(\omega) \wedge \sigma_0(\Omega) \wedge \sigma_0(\Omega)$

```
Clear[ZeroMinusOneTerm1, ZeroMinusOneTerm2, ZeroMinusOneTerm3]
```

Definition of the 5-form Tr $\sigma_{-1}(\omega) \wedge \sigma_0(\Omega) \wedge \sigma_0(\Omega)$ using the expressions above for the symbols. The 1/120 factor is added after the evaluation of the pullback.

```
ZeroMinusOneTerm1[TF_, WF_, XF_, YF_, ZF_] := Sum[SigmaMinusConn[TF][[mm, jj]] *
   SigmaZeroCur[WF, XF][[jj, ii]] * SigmaZeroCur[YF, ZF][[ii, mm]],
  {ii, 1, dimension}, {jj, 1, dimension}, {mm, 1, dimension}]
```

Compute the evaluation of every term of Tr $\sigma_{-1}(\omega) \wedge \sigma_0(\Omega) \wedge \sigma_0(\Omega)$ on all possible permutations of T, X, Y, W, Z. The result is stored in the matrix Final.

```
Final = Apply[ZeroMinusOneTerm1, Permu, {1}];
```

```
DotGam = << GammaDot5.dat;
XFF = << S1xS2XField.dat;
YFF = << S1xS2YField.dat;
WFF = << S1xS2ZField.dat;
```

The variable Final is a vector with 120 entries. We add all the entries with the corresponding sign for the permutation.

```
Fin = Sum[Final[[j]] * SignPermu[[j]], {j, 1, Length[SignPermu]}];
```

Simplification of the expression for general a. Notice that the contribution vanishes when a = 1.

```
TotalComputation = Simplify[Fin]
```

$$\frac{32\,(-1+a)^2\,\text{PAlpha}\;y\;\text{Sin}[\text{Theta}]}{3\,(-1+y)^5}$$

Multiplication by the 1/120 factor.

```
TotalComputation = TotalComputation * 1 / 120
```

$$\frac{4\,(-1+a)^2\,\text{PAlpha}\;y\;\text{Sin}[\text{Theta}]}{45\,(-1+y)^5}$$

We have to integrate the variableTotalComputation over $S^2 \times S^3$. We begin with the integration with respect to the variable $\theta$.

```
TotalComputationOne = Integrate[TotalComputation, {Theta, 0, Pi}]
```

$$\frac{8\,(-1+a)^2\,\text{PAlpha}\,y}{45\,(-1+y)^5}$$

Now, we integrate with respect to y. We do this in two steps. First we compute the indefinite integral.

```
FunctionSigma = TotalComputationOne;
FunctionSigma = Simplify[Integrate[FunctionSigma, y]]
```

$$\frac{2\,(-1+a)^2\,\text{PAlpha}\,(1-4\,y)}{135\,(-1+y)^4}$$

For the second step, we evaluate the integral at the bounds $y_1 = 1/2\left(1-\lambda-\text{Sqrt}\left[1-\lambda^2/3\right]\right)$ and $y_2 = y_1 + \lambda$. We also express the value of $a = 3\,y_1^2 - 2\,y_1^3$ in terms of $\lambda$ as in (3.3), (3.4) and (3.5) in Gauntlett et al.

```
TotalComputationFunction[y_] := 2 (-1 + a)^2 PAlpha (1 - 4 y) / (135 (-1 + y)^4)
```

$$\text{TotalComputationFunction}[y\_] := \frac{2\,(-1+a)^2\,\text{PAlpha}\,(1-4\,y)}{135\,(-1+y)^4}$$

```
Rooty1 = 1 / 2 (1 - l - Sqrt[1 - l^2 / 3]);
Rooty2 = Rooty1 + l;
```

```
FunctionPlot =
 Simplify[TotalComputationFunction[Rooty2] - TotalComputationFunction[Rooty1] /.
    a -> 3 * Rooty1^2 - 2 * Rooty2^3] /. l -> λ
```

$$\frac{1}{45\left(3-2\lambda^2+\sqrt{9-3\lambda^2}\right)^4}8\,\text{PAlpha}\,\lambda^3\left(144\lambda^5\sqrt{9-3\lambda^2}+252\lambda^4\left(-5+\sqrt{9-3\lambda^2}\right)-108\left(3+\sqrt{9-3\lambda^2}\right)-108\lambda^3\left(3+\sqrt{9-3\lambda^2}\right)+16\lambda^6\left(3+\sqrt{9-3\lambda^2}\right)+9\lambda^2\left(57+17\sqrt{9-3\lambda^2}\right)\right)$$

As in (3.1), (3.8) in Gauntlett et al., we set \lambda = 3 q/(2 p) for fixed relatively prime integers p, q satisfying 4 p^2 – 3 q^2 = n^2 for a positive integer n.

$$\text{FunctionPlot} = \text{Simplify}\left[\text{FunctionPlot} \;/.\; \left\{\sqrt{9-3\lambda^2} \to (3\,n)/(2\,p),\ \lambda \to (3\,q)/(2\,p)\right\}\right]$$

$$\frac{1}{5p^2\left(n p+2p^2-3q^2\right)^4}6\,\text{PAlpha}\,q^3\left(-2p\left(16p^6-57p^4q^2+54p^3q^3+315p^2q^4-27q^6\right)+n\left(-16p^6+51p^4q^2-54p^3q^3+189p^2q^4+162p q^5+27q^6\right)\right)$$

To compute the total contribution we trivially integrate with respect to $0 < \alpha < 2\pi$ l,
where l = q/(3 q^2 – 2 p^2 + p (4 p^2 – 3 q^2)^(1/2)) as in (3.6) in Gauntlett et al.;
with respect to $\phi$, which gives a factor of $2\pi$; with respect to $\psi$, which gives a factor of $\pi$,
and with respect to the variables of the unit cotangent bundle of the loop,
which gives a factor of $(2)(2\pi) = 4\pi$. This shows that the evaluation is a rational number times some powers of $\pi$. The following function allows to compute the total contribution plugging only values for p and q.

In[70]:=
$$\text{TotalContribution}[p\_, q\_] := \text{Module}\Big[\{n\},\ n = \text{Sqrt}[4\,p^{\wedge}2 - 3\,q^{\wedge}2];$$
$$\frac{1}{5p^2\left(n p+2p^2-3q^2\right)^4}6\,\text{PAlpha}\,q^3\left(-2p\left(16p^6-57p^4q^2+54p^3q^3+315p^2q^4-27q^6\right)+n\left(-16p^6+51p^4q^2-54p^3q^3+189p^2q^4+162p q^5+27q^6\right)\right)*$$
$$2^{\wedge}4 * \text{Pi}^{\wedge}4 * (q/(3\,q^{\wedge}2 - 2\,p^{\wedge}2 + p*n))\Big]$$

We compute for (p, q ) = (7, 3) and for (p, q) = (19, 5), corresponding to the values (A, B) = (3, 1) and (A, B) = (5, 1), respectively, as explained in (3.12) in Gauntlett et al. The results are nonzero.

In[74]:= **TotalContribution[7, 3]**

Out[74]= $-\frac{1849\ \text{PAlpha}\ \pi^4}{22\,050}$

In[73]:= **TotalContribution[19, 5]**

Out[73]= $-\frac{312\,481\ \text{PAlpha}\ \pi^4}{36\,843\,660}$

Conclusion: We compute that the WCS integral over $S^2 \times S^3$ for the metric described in Gauntlett et al. is nonzero.